

FIRST CANADIAN COMPUTER NEWS



VOL. 1 NO. 4 SUMMER 1977

Winners:

Judging the entries in the First Canadian Computer Contest was a real delight. The winners are :

First Prize : \$300 : Wayne Hammerschlag : Battleship
Second Prize : \$150 : Duncan Elliott : Master Mind
Third Prize : \$ 50 : Chris Gray : First Canadian Car Race

We were surprized to learn that Wayne and Duncan are High School students (Ph.D's eat your hearts out), and even more surprized to discover that they both attend Silverthorn Collegiate Inst. in Etobicoke. So our congratulations also go to their teacher, Mr. Lovelace, who is obviously doing an excellent job.

Wayne's program was selected because, in addition to being a very interesting game to play, it is excellently documented. In fact, many manufacturers supplying the hobby computer market would do well to follow this example.

Duncan's program is a gem. It uses only 50 lines of code, but it keeps you going for hours at any level of difficulty you wish to set for yourself.

Chris's program has all that a good game should have. It is simple, easy to learn, fast and challenging.

As we promised, the winning programs are reprinted in this newsletter. Wayne's and Duncan's programs were both written for the PDP-8. Chris's program was written for the SWTP 6800. These may require minor modifications if you wish to run them on other systems.

SOLO owners tune up !

Software Technology's MUSIC SYSTEM is available in a package that runs in 2K and includes :


- . A high level music language with a one pass compiler with output that can be located anywhere in memory. Any 3-part score can be entered. Even non-musicians can use it with ease.
- . All standard musical notation is supported, including any key signature : any time signature : any clef notation : note values from whole to 1/64th. notes : rests : dotted notes : triplets : staccato : articulation : accidentals.
- . Full support for repeats : second endings : refrains.
- . Transpose to or from any key.

All you need is your own hi-fi (with amplifier) and jack.

Complete hardware and software package available from stock.

\$30.00 postpaid.

BATTLESHIP



NAVAL INTELLIGENCE HAS JUST LEARNED THE APPROXIMATE LOCATION OF THE GERMAN BATTLESHIP BISMARCK WHICH IS A MENACE TO CARGO SHIPS IN YOUR AREA. ON THE FIRING GRID PRESENTED TO YOU LIES THE BISMARCK, HE CANNOT MOVE BEYOND IT. NATURALLY HE WILL BE MOVING CONSTANTLY AT RANDOM SPEED AND DIRECTION BUT THE BISMARCK, AS WELL AS YOUR OWN VESSEL, CAN OCCUPY ONLY ONE GRID UNIT AT A TIME. A SHOT ON THIS GRID UNIT WILL BE A MAJOR HIT WHILE A SHOT WITHIN 5 GRID UNITS WILL BE A MINOR HIT. ANY HIT ON THE BISMARCK WILL REDUCE ITS SPEED AND FIRING RANGE, YES HE CAN AND WILL FIRE BACK AT YOU WHEN IN RANGE. YOU TO HAVE A LIMITED FIRING RANGE SO BEFORE YOU CAN EVEN DAMAGE THE BISMARCK YOU MUST MOVE WITHIN RANGE. THE BISMARCK WILL TAKE EVASIVE MANOEUVERS WHEN IT SUFFERS A MINOR HIT OR SENSES YOUR POSITION (I.E. HE WILL MOVE IN THE OPPOSITE DIRECTION WHEN WITHIN 5 GRID UNITS FROM YOU OR AFTER A MINOR HIT). YOUR VESSEL HAS A LIMITED FUEL AND POWER SUPPLY, LIMITED AMMUNITION, LIMITED FIRING RANGE AND A LIMITED MAXIMUM SPEED. YOU HAVE A CHOICE OF ONE OF FOUR COMMANDS:

#1: STATUS REPORT

USES 2 UNITS OF POWER
TELLS YOUR ;
POWER LEVEL
AMMUNITION LEVEL
GRID POSITION
MAXIMUM SPEED
MAXIMUM GUN RANGE
AND OPTIONAL DAMAGE REPORTS

#2: SONAR

USES 5 UNITS OF POWER
GIVES LAST REPORTED DISTANCE OF BISMARCK WITH DIRECTION
(IF YOU ARE WITHIN RANGE BISMARCK WILL SENSE YOUR SONAR PULSES AND FIRE AT YOU)

#3: COURSE CHANGE

USES 1 POWER UNIT
ALLOWS CHANGE OF YOUR HEADING
INPUT TO BE ENTERED IN THE FOLLOWING MANNER;

"COMMAND" COURSE [RETURN]
"NEW HEADING" WEST FULL [RETURN]

WHERE WEST IS THE DIRECTION YOU WISH TO TRAVEL AND FULL IS THE SPEED (IN THIS CASE MAX. BUT YOU MAY ENTER AN INTEGER VALUE BETWEEN 1 AND THE MAX GRID UNIT SPEED - ATTAINED BY DIVIDING YOUR NAUTICAL SPEED GIVEN ON STATUS REPORT BY 5)

* NOTE: TO STOP YOUR VESSEL REPLY TO "NEW COURSE" WITH "FULL STOP" AND YOUR VESSEL WILL REMAIN STATIONARY, OTHERWISE YOUR SHIP WILL BE IN CONSTANT MOTION WITH DIRECTION AND SPEED DETERMINED BY YOUR COURSE HEADING SPECIFIED. YOUR POSITION CAN EASILY BE CALCULATED FROM YOUR COURSE HEADING SPECIFICATIONS AND YOUR STATUS REPORT. YOUR VESSEL MOVES ONE GRID SPACE PER GRID UNIT SPEED EVERY COMMAND CYCLE (I.E. EVERY TIME "COMMAND" IS PRINTED ON THE TERMINAL)

E.G.

IF YOUR COURSE WAS WEST 5 AND YOUR STATUS REPORT STATED YOUR WERE AT 30,40 WHEN YOU ENTER YOUR NEXT COMMAND YOUR VESSEL WILL BE AT 25,40.

#4: SHOOT

USES 1 UNIT OF POWER
FIRES ONE SHOT
INPUT TO BE ENTERED IN THE FOLLOWING MANNER;
[COMMAND]?SHOOT X,Y

WHERE X,Y REPRESENT THE COORDINATES OF THE SHOT IN A CARTESIAN GRID WITH YOUR VESSEL AS THE ORIGIN. E.G.

SHOOT 6,-4 WOULD FIRE A SHOT 6 GRID UNITS TO THE RIGHT OF YOUR VESSEL AND 4 UNITS BELOW.

40 REM
45 REM
50 REM
55 REM
60 REM
65 REM
70 REM
75 REM
80 REM
85 REM
90 REM
95 REM

THIS PROGRAM WAS WRITTEN FOR THE FIRST CANADIAN COMPUTER STORE LTD. AS AN ENTRY TO THE FIRST CANADIAN COMPUTER CONTEST.....

WRITTEN BY :

MR. WAYNE HAMMERSCHLAG
7 FIELDSTONE RD. ETOBICOKE
ONTARIO, CANADA M9C-2J5

«««««««« »»»»»»»»
«««« BATTLE SHIP »»»»
«««««««« »»»»»»»»

VARIABLE DIRECTORY

=====

A# = INPUT COMMAND BUFFER
S = SONAR RANGE (GRID UNITS)
S1 = BISMARCK'S MAX. SPEED (GRID UNITS)
S2 = PLAYER'S MAX. SPEED
A1 = BISMARCK'S AMMUNITION LEVEL
A2 = PLAYER'S AMMUNITION LEVEL
D1 = BISMARCK'S DAMAGE LEVEL
D2 = PLAYER'S DAMAGE LEVEL
P = POWER LEVEL
R1 = BISMARCK'S MAX. GUNNING RANGE (GRID UNITS)
R2 = PLAYER'S MAX. GUNNING RANGE (GRID UNITS)
G1 = BISMARCK'S GUNNING ACCURACY (MINOR HIT DIF.)
G2 = PLAYER'S GUNNING ACCURACY (MINOR HIT DIF.)
B = GRID SIZE
C1 = PLAYER'S COURSE DIRECTION
C2 = PLAYER'S COURSE SPEED
L1,L2 = BISMARCK'S GRID COORDINATES (X,Y)
L3,L4 = PLAYER'S GRID COORDINATES (X,Y)

INITIALIZATION

=====

400 DIM A\$(72)
410 PRINT "DO YOU WISH BRIEFING";
420 INPUT A\$
430 IF SEG\$(A\$,1,1) = "Y" THEN 490
440 PRINT "DO YOU WISH GRID MAP";
450 INPUT A\$
460 IF SEG\$(A\$,1,1) = "N" THEN 500
470 GOSUB 3710
480 GOTO 500
490 GOSUB 3630
500 RANDOMIZE
510 LET S=15\S1=3\S2=5\A1=50\A2=40\D1=10\D2=10\P=150\R1=8\R2=6
520 LET G1=3\G2=5\B=50\C1=0\C2=0
530 LET X = INT (RND(0)*B+ 5)
540 LET Y = INT (RND(0)*B+ 5)
550 IF X=25 THEN 530
560 IF Y=25 THEN 540
570 LET L1=X\L3=B-X
580 LET L2=Y\L4=B-Y
590 LET V=SGN(X-(B/2))*-1\H=SGN(Y-(B/2))*-1
600 GOSUB 2230
610 REM
620 REM COMMAND DECODER
630 REM =====
640 REM
650 GOSUB 2680
660 GOSUB 3460
670 IF P <= 0 THEN 1470
680 PRINT "COMMAND";
690 INPUT A\$
700 IF SEG\$(A\$,1,6) = "STATUS" THEN 1550
710 IF SEG\$(A\$,1,5) = "SHOOT" THEN 800
720 IF SEG\$(A\$,1,5) = "SONAR" THEN 1260
730 IF SEG\$(A\$,1,6) = "COURSE" THEN 3080
740 PRINT "ILLEGAL COMMAND !"
750 GOTO 680
760 REM
770 REM FIRING ROUTINE
780 REM =====
790 REM
800 IF A2 = 0 THEN 1610
810 LET P =P-1
820 LET X=VAL(SEG\$(A\$,7,POS(A\$,",",5))-1)
830 IF ABS(X) <= R2 THEN 870
840 PRINT "ERROR IN SHOT COORDINATES"
850 PRINT "OUT OF OUR RANGE , CHECK STATUS !!!"
860 GOTO 650
870 LET X=X+L3
880 LET Y=VAL(SEG\$(A\$,POS(A\$,",",5)+1,LEN(A\$)))
890 IF ABS(Y) <= R2 THEN 910
900 GOTO 840
910 LET Y=Y+L4
920 LET A2=A2-1
930 IF L1 <> X THEN 950
940 IF L2=Y THEN 1090
950 IF ABS(L2-Y) > G2 THEN 1180
960 IF ABS(L1-X) > G2 THEN 1180
970 PRINT "A MINOR HIT ON THE BISMARK!"
980 LET D1=D1-1
990 LET V=SGN(L2-Y)
1000 LET H=SGN(L1-X)
1010 IF S1 = 0 THEN 1040
1020 LET S1 = S1-1
1030 GOSUB 1960
1040 GOSUB 2230
1050 LET X1=S1
1060 LET Y1=S1
1070 GOSUB 2820
1080 GOTO 650
1090 PRINT "A MAJOR HIT ON THE BISMARK !!!"
1100 LET D1 = D1-3
1110 LET S1 = S1-2
1120 IF S1 > 0 THEN 1150
1130 LET S1 = 0

```

1140 GOSUB 1960
1150 GOSUB 2230
1160 GOSUB 2680
1170 GOTO 650
1180 PRINT "A MISS !"
1190 GOSUB 2230
1200 GOSUB 2680
1210 GOTO 650
1220 REM
1230 REM          SONAR ROUTINE
1240 REM          =====
1250 REM
1260 LET P = P-5
1270 IF ABS(L1-L3) > S THEN 1450
1280 IF ABS(L2-L4) > S THEN 1450
1290 IF S1 <= 0 THEN 1420
1300 PRINT " BISMARK'S SPEED =" ; (X1+Y1)*3.5 ; "KNOTS"
1310 PRINT "          DISTANCE =" ; ABS(L1-L3) ; " ; " ; ABS(L2-L4)
1320 PRINT "          DIRECTION =" ;
1330 IF SGN(H) = -1 THEN 1360
1340 PRINT " NORTH" ;
1350 GOTO 1370
1360 PRINT " SOUTH" ;
1370 IF SGN(V) = -1 THEN 1400
1380 PRINT " WEST"
1390 GOTO 1440
1400 PRINT " EAST"
1410 GOTO 1440
1420 PRINT " BISMARK'S ENGINES OUT"
1430 PRINT " CAN'T LOCATE !!!"
1440 GOTO 600
1450 PRINT " NOTHING IN RANGE"
1460 GOTO 660
1470 PRINT " OUT OF FUEL & POWER !!!!"
1480 STOP
1490 PRINT " OUT OF AMMUNITION !!!!"
1500 STOP
1510 REM
1520 REM          STATUS ROUTINE
1530 REM          =====
1540 REM
1550 LET P=P-2
1560 IF P=> 0 THEN 1580
1570 LET P=0
1580 PRINT " AMMUNITION LEVEL =" ; A2 ; " SHOTS LEFT"
1590 PRINT " MAXIMUM GUN RANGE =" ; INT(R2) ; " UNITS"
1600 PRINT " YOUR POSITION IS " ; L3 ; " ; " ; L4
1610 PRINT " POWER & FUEL LEVEL =" ; P ; " UNITS"
1620 IF D2 = 10 THEN 1760
1630 IF D2 > 9 THEN 1760
1640 IF D2 < 5 THEN 1710
1650 PRINT " MINOR DAMAGE TO AFT , DECK FIRE UNDER CONTROL"
1660 IF D2 > 8 THEN 1690
1670 PRINT " 2 MEN KILLED , 7 BADLY INJURED"
1680 GOTO 1760
1690 PRINT " 5 MEN BADLY INJURED"
1700 GOTO 1760
1710 PRINT " HEAVY DAMAGE TO AFT , PORT MAIN GUNS DISABLED"
1720 IF D2 < 3 THEN 1750
1730 PRINT " 7 MEN KILLED , 17 BADLY INJURED"
1740 GOTO 1760
1750 PRINT " 12 MEN KILLED , 28 BADLY INJURED"
1760 IF S2 < 3 THEN 1820
1770 IF S2 < 5 THEN 1800
1780 PRINT " SHIP'S ENGINES IN NORMAL CONDITION, TOP SPEED 35 KNOTS"
1790 GOTO 1900
1800 PRINT " SHIP'S ENGINES SUFFERED MINOR DAMAGE, TOP SPEED" ; S2*7 ; " KNOTS"
1810 GOTO 1900
1820 IF S2 > 0 THEN 1850
1830 PRINT " SHIP'S ENGINES DAMAGED BEYOND REPAIR , CAN'T MANOEUVRE"
1840 GOTO 1900
1850 PRINT " SHIP'S ENGINES SEVERELY DAMAGED , TOP SPEED" ; S2*7 ; " KNOTS"
1860 IF P > 0 THEN 1900
1870 IF A2 > 0 THEN 1900
1880 PRINT " WE'RE DONE FOR ! ABANDON SHIP"
1890 STOP
1900 IF D1 > 0 THEN 650
1910 STOP
1920 REM
1930 REM          BISMARK SURVEYED STATUS ROUTINE
1940 REM          =====
1950 REM
1960 IF D1 < 3 THEN 2020
1970 IF D1 = 10 THEN 2060
1980 IF D1 > 6 THEN 2040
1990 PRINT " SEVERE DAMAGE, VISIBLE DECK FIRE"
2000 LET R1= R1-(10-D1)/7
2010 GOTO 2060
2020 PRINT " PARTIALLY CRIPPLED, MAIN GUNS OUT , SIGNED FOR HELP"
2030 GOTO 2000
2040 PRINT " MINOR DAMAGE"
2050 GOTO 2000
2060 IF S1 < 1 THEN 2150
2070 IF S1 < 2 THEN 2130
2080 IF S1 < 3 THEN 2110
2090 PRINT " ENGINES IN GOOD CONDITION"
2100 RETURN
2110 PRINT " MINOR ENGINE DAMAGE"
2120 RETURN
2130 PRINT " ENGINE'S BADLY DAMAGED"
2140 RETURN
2150 PRINT " ENGINES OUT , CAN'T MANOEUVRE !!!"
2160 IF D1 > 0 THEN 2140
2170 PRINT " IT'S SINKING!!! -- YOU'VE DONE IT !! "
2180 GOTO 1550
2190 REM
2200 REM          BISMARK SHOT ROUTINE
2210 REM          =====
2220 REM
2230 IF A1 <= 0 THEN 2580
2240 IF ABS(L1-L3) > R1+G1 THEN 2350
2250 IF ABS(L2-L4) > R1+G1 THEN 2350
2260 LET X = INT(RND(O)*R1+ .5)*SGN(L3-L1)
2270 LET Y = INT(RND(O)*R1+ .5)*SGN(L4-L2)
2280 IF X<< 0 THEN 2360
2290 IF Y<<0 THEN 2360
2300 LET D1=D1-2
2310 LET A1=A1-2
2320 PRINT " ACCIDENTAL PREMATURE EXPLOSION ON BOARD BISMARK !!!"
2330 GOSUB 1960
2340 GOSUB 2680
2350 RETURN
2360 LET X=L1+X
2370 LET A1=A1-1
2380 LET Y=L2+Y
2390 IF X >< L3 THEN 2410
2400 IF Y = L4 THEN 2520
2410 IF ABS(L3-X) < G1 THEN 2440
2420 PRINT " HE MISSED"
2430 RETURN
2440 IF ABS(L4-Y) > G1 THEN 2420
2450 PRINT " YOU SUFFERED A MINOR HIT!"
2460 LET D2=D2-1
2470 LET R2=R2-(10-D2)/5
2480 IF D2>0 THEN 2500
2490 PRINT " YOU'RE SINKING !!!!" \ GOTO 1880
2500 LET S2=S2-1
2510 RETURN
2520 PRINT " YOU SUFFERED A MAJOR HIT !"
2530 LET D2=D2-2.5
2540 LET R2=R2-(10-D2)/5
2550 LET S2=S2-2
2560 IF D2 <= 0 THEN 2490
2570 RETURN
2580 IF K = 1 THEN 2570
2590 LET K=1
2600 PRINT " RADIO INTERCEPT>>>>>BISMARK OUT OF AMMUNITION !!!"
2610 RETURN
2620 REM
2630 REM          BISMARK MOVE ROUTINE
2640 REM          =====
2650 REM
2660 REM          RANDOM COURSE ENTRY
2670 REM
2680 IF ABS(L1-L3) > 5 THEN 2750
2690 IF ABS(L2-L4) > 5 THEN 2750
2700 LET X1 = S1
2710 LET H = SGN(L1-L3)
2720 LET Y1 =S1
2730 LET V = SGN(L2-L4)
2740 GOTO 2820
2750 LET H = SGN(SGN(RND(O)*2-1)+.1)
2760 LET X1 = INT(RND(O)*S1+ .5)
2770 LET V=SGN(SGN(RND(O)*2-1)+.1)
2780 LET Y1 = INT(RND(O)*S1+ .5)
2790 REM
2800 REM          EVASIVE MANOEUVRE ENTRY
2810 REM
2820 LET L1 = ABS(L1+(X1*H))
2830 LET L2 = ABS(L2+(Y1*V))
2840 FOR U1 = 1 TO X1
2850   IF L3 = ABS(L1-(U1*H)) THEN 2880
2860   IF L3 = 2*B-(L1-(U1*H)) THEN 2880
2870 NEXT U1
2880 GOTO 2930
2890 FOR U1 = 1 TO Y1
2900   IF L4 = ABS(L2-(U1*V)) THEN 3010
2910   IF L4 = 2*B-(L2-(U1*V)) THEN 3010
2920 NEXT U1
2930 IF L1<= B THEN 2960
2940 LET L1=2*B-L1
2950 LET H=H*-1
2960 IF L2<= B THEN 2990
2970 LET L2=2*B-L2
2980 LET V=V*-1
2990 IF L1>L3 THEN 3030
3000 IF L2>L4 THEN 3030
3010 PRINT " THE BISMARK RAMMED YOU !!!"
3020 GOTO 1880
3030 RETURN
3040 REM
3050 REM          COURSE CHANGE
3060 REM          =====
3070 REM
3080 PRINT " NEW COURSE" ;
3090 LET P = P-1
3100 INPUT A#
3110 IF SEG$(A#,1,9) = "FULL STOP" THEN 3340
3120 IF SEG$(A#,1,5) = "NORTH" THEN 3300
3130 IF SEG$(A#,1,5) = "SOUTH" THEN 3260
3140 IF SEG$(A#,1,4) = "EAST" THEN 3220
3150 IF SEG$(A#,1,4) = "WEST" THEN 3180
3160 PRINT " ARE YOU NUTS ??????"
3170 GOTO 650
3180 LET B# = SEG$(A#,6,LEN(A#))
3190 GOSUB 3360
3200 LET C1=4
3210 GOTO 650
3220 LET B# = SEG$(A#,6,LEN(A#))
3230 GOSUB 3360
3240 LET C1=2
3250 GOTO 650
3260 LET B# = SEG$(A#,7,LEN(A#))
3270 GOSUB 3360
3280 LET C1=3
3290 GOTO 650
3300 LET B# = SEG$(A#,7,LEN(A#))
3310 GOSUB 3360
3320 LET C1=1
3330 GOTO 650
3340 LET C2=0:NC1=0
3350 GOTO 650
3360 IF SEG$(B#,1,4)="FULL" THEN 3400
3370 LET C2 = VAL(SEG$(B#,1,LEN(B#)))
3380 IF C2 > S2 THEN 3440
3390 RETURN
3400 LET C2=S2
3410 RETURN

```

BTLSHP. BA

PAGE 1

```

3420 PRINT "WE COLLIDED WITH THE BISMARK !"
3430 GOTO 1880
3440 PRINT "CHECK STATUS , CAN'T MAKE THAT SPEED !"
3450 GOTO 650
3460 IF C1/2 = INT(C1/2) THEN 3530
3470 LET L4 = L4-(SGN(C1-2.5)*C2)
3480 IF L3<C L1 THEN 3520
3490 FOR U1 = 1 TO C2
3500   IF L2 = L4+(SGN(C1-2.5)*U1) THEN 3420
3510 NEXT U1
3520 RETURN
3530 LET L3 = L3-(SGN(C1-2.5)*C2)
3540 IF L4 >C L2 THEN 3580
3550 FOR U1 = 1 TO C2
3560   IF L1 = L3+(SGN(C1-2.5)*U1) THEN 3420
3570 NEXT U1
3580 RETURN
3590 REM
3600 REM      INSTRUCTION AND MAPPING ROUTINE
3610 REM      =====
3620 REM
3630 FILE#1:"BRIEF.LS"
3640 FILE#2:"LPT:"
3650 INPUT#1:A#
3660 IF END#1 THEN 3690
3670 PRINT#2:A#
3680 GOTO 3650
3690 CLOSE#1
3700 CLOSE#2
3710 FILE#1:"LPT:"
3720 PRINT#1:\PRINT#1:\PRINT#1:
3730 GOSUB 3960
3740 PRINT#1:
3750 PRINT#1:TAB(7);50;
3760 GOSUB 4000
3770 PRINT#1:50
3780 FOR Y = 49 TO 1 STEP -1
3790   IF Y/5 >C INT(Y/5) THEN 3840
3800 PRINT#1:TAB(7);Y;
3810 GOSUB 4000
3820 PRINT#1:Y
3830 GOTO 3880
3840 FOR X = 11 TO 61 STEP 5
3850   PRINT#1:TAB(X);"+";
3860 NEXT X
3870 PRINT#1:
3880 NEXT Y
3890 PRINT#1:TAB(7);0;
3900 GOSUB 4000
3910 PRINT#1:0
3920 GOSUB 3960
3930 PRINT#1:
3940 CLOSE#1
3950 RETURN
3960 FOR X = 0 TO 50 STEP 5
3970 PRINT#1:TAB(X+10);X;
3980 NEXT X
3990 RETURN
4000 FOR X = 0 TO 50
4010 PRINT#1:TAB(X+11);"+";
4020 NEXT X
4030 RETURN
4040 END

```

BRANCH	FROM LINES
4000	3760\$, 3810\$, 3900\$
3960	3730\$, 3920\$
3880	3830#
3840	3790*
3710	470\$
3690	3660*
3650	3680#
3630	490\$
3580	3540*
3530	3460*
3520	3480*
3460	660\$
3440	3380*
3420	3560*, 3500*
3400	3360*
3360	3190\$, 3230\$, 3270\$, 3310\$
3340	3110*
3300	3120*
3260	3130*
3220	3140*
3180	3150*
3080	730*
3030	3000*, 2990*
3010	2910*, 2900*
2990	2960*
2960	2930*
2880	2860*, 2850*
2820	2740#, 1070\$
2750	2680*, 2690*
2680	650\$, 1160\$, 1200\$, 2340\$
2580	2230*
2570	2580*
2520	2400*
2500	2480*
2490	2560*
2440	2410*
2420	2440*
2410	2390*
2360	2280*, 2290*
2350	2240*, 2250*
2230	1040\$, 1190\$, 600\$, 1150\$
2150	2060*
2140	2160*
2130	2070*
2110	2080*
2060	2010#, 1970*
2040	1980*
2020	1960*
2000	2050#, 2030#
1960	2330\$, 1140\$, 1030\$
1900	1790#, 1810#, 1840#, 1860*, 1870*
1880	3430#, 3020#, 2490#
1850	1820*
1820	1760*
1800	1770*
1760	1630*, 1680#, 1700#, 1740#, 1620*
1750	1720*
1710	1640*
1690	1660*
1610	800*
1580	1560*
1550	2180#, 700*
1470	670*
1450	1280*, 1270*
1440	1410#, 1390#
1420	1290*
1400	1370*
1370	1350#
1360	1330*
1260	720*
1180	960*, 950*
1150	1120*
1090	940*
1040	1010*
950	930*
910	890*
870	830*
840	900#
800	710*
680	750#
660	1460#
650	3250#, 1080#, 3290#, 1210#, 3330#
	3350#, 1170#, 3170#, 860#, 3450#, 3210#
	1900*
600	1440#
540	560*
530	550*
500	480#, 460*
490	430*

```

# = GOTO
* = IF-THEN
$ = GOSUB

```

PDF Created by David Crawford, www.the-crawfords.com

MASTER MIND

```
10 PRINT"          MASTER MIND"
20 REM DUNCAN ELLIOTT IF MAY 10
30 DIM C(30),D(30),E(30)
40 DIM P$(1)
50 PRINT "HOW MANY CHARACTERS";
60 INPUT A
70 PRINT "HOW MANY VALUES PER CHARACTER";
80 INPUT B
90 PRINT "BY HOW MUCH CAN THE CHARACTERS INCREASE";
100 INPUT R
110 PRINT
120 FOR F=1 TO A
130 RANDOMIZE
140 LET C(F)=INT(RND(F)*B+1)
150 NEXT F
160 LET Q=INT(RND(O)*R)
170 FOR H=1 TO 50
180 FOR G=1 TO A
190 LET D(G)=C(G)+H*Q
200 NEXT G
210 PRINT
220 PRINT "ENTER YOUR GUESS"
230 FOR I=1 TO A
240 INPUT E(I)
250 NEXT I
260 LET J=0
270 PRINT
280 FOR K=1 TO A
290 IF D(K)<>E(K) GOTO 330
300 LET J=J+1
310 LET D(K)=-1
320 LET E(K)=-2
330 NEXT K
340 LET L=0
350 FOR M=1 TO A
360 FOR N=1 TO A
370 IF D(M)<>E(N) GOTO 410
380 LET L=L+1
390 LET D(M)=-1
400 LET E(N)=-2
410 NEXT N
420 NEXT M
430 PRINT J;"HITS",
440 PRINT L;"ARE CLOSE"
450 IF J=A GOTO 470
460 NEXT H
470 PRINT "YOU'RE FINISHED IN";H;"MOVES"
480 PRINT "ANOTHER GAME (Y,N)";
490 INPUT P$
500 IF P$="Y" THEN 110
510 END
```

This program is a variation of MASTERMIND (TM).

For each game, the player indicates how many characters (pins) he wishes to play with, and the number of values which each character may have. All characters will have the same number of values, but each will have its own value !!!

In addition, the player may give a value, based on which the computer selects a random number (integer, between zero and the value given) and will use this to increment the character values on each step. For any game, the chosen number remains constant and is added to all characters on every turn.

FIRST CANADIAN CAR RACE

```
10 PRINT CHR(16),CHR(22)
20 PRINT "THE FIRST CANADIAN CAR RACE"
30 PRINT
40 PRINT
50 PRINT "IF YOU WANT INSTRUCTIONS ENTER '0'"
60 PRINT "TO START THE RACE ENTER '1'"
70 INPUT A
80 IF A=0 GOSUB 650
90 DATA 900,80,1100,50,500,40
100 DATA 600,45,1000,70,1700,0
110 READ A1, A2
120 A3=3*A1
130 IF A3<0 THEN A3=0
140 PRINT INT(A3/3);"YDS ";INT(V*15/22);" MPH<";A2;" ";R;"RPM"
150 INPUT G
160 T=T+2
170 IF G=9 GOTO 220
180 IF C*G>25 GOTO 150
190 IF G>0 GOTO 270
200 IF G<0 GOTO 300
210 IF G=0 GOTO 240
220 T=T-2
230 GOTO 130
240 S=V*2
250 G=G0
260 GOTO 360
270 A=18-3*G
280 G0=G
290 GOTO 320
300 A=3*G
310 G0=G0-G/2
320 G1=G0-1
330 U=V
340 V=U+A+A
350 S=(V*V-U*U)/2/A
360 IF S<=0 GOTO 520
370 IF S<A3 GOTO 450
380 W=INT(V*15/22)
390 IF W>A2 GOTO 550
400 N=N+1
410 IF N=5 GOTO 620
420 IF N<5 GOTO 110
430 PRINT T;" SECS"
440 END
450 A3=A3-S
460 R=80*V
470 FOR I=1 TO G1
480 R=INT(R*.75)
490 NEXT I
500 IF R>6000 GOTO 590
510 GOTO 130
520 PRINT "THIS IS A PIT STOP !!!"
530 V=0
540 GOTO 130
550 IF A2=0 GOTO 430
560 PRINT "CRASHED !!! LOSE 10 SECS"
570 T=T+10
580 GOTO 450
590 PRINT "WATCH THOSE REVS !! LOSE 3 SECS"
600 T=T+3
610 GOTO 130
620 PRINT CHR(16),CHR(22)
630 PRINT "LAST STRETCH... MOVE IT !!!"
640 GOTO 110
650 PRINT CHR(16),CHR(22)
660 PRINT"THE FIRST CANADIAN CAR RACE"
670 PRINT"EVERY 2 SECS (CAR TIME), YOU WILL"
680 PRINT"BE SHOWN DISTANCE TO NEXT CURVE,"
690 PRINT"YOUR PRESENT SPEED, MAXIMUM"
700 PRINT"CURVE SPEED AHEAD, YOUR PRESENT"
710 PRINT"RPM (MEASURES ENGINE SPEED, A"
720 PRINT"FUNCTION OF CAR SPEED + GEAR)."
730 PRINT" 1(MOST) TO 5 ACCELERATES CAR"
740 PRINT" -1 TO-5(MOST) SLOWS CAR"
750 PRINT"0 MAINTAINS STEADY CAR SPEED"
760 PRINT" 9 RE-DISPLAYS CURRENT STATUS"
770 PRINT"YOU WILL BE PENALISED IF CURVE"
780 PRINT"SPEED IS TOO HIGH OR IF RPM"
790 PRINT"IS OVER 6000. WHEN READY HIT '1'"
800 INPUT A
810 RETURN"
```

8000 Multitasking.

8000MR is a multitasking system by Microware Systems Corporation, a direct replacement for Mikbug(TM) in 6800 systems.

of operation are provided in RT/68MR :

Mode provides ten command functions for entering, changing memory and register contents, plus executing breakpoints.

allows execution of programs that were written for systems that cannot run as tasks.

allows CPU time to be shared by up to 16 tasks (programs), thus allowing concurrent operation of several programs. Tasks may be dependent or independent, and may control execution of each other, and may communicate by means of flags or common data areas. In addition, real time processing allows tasks to be time dependent or to respond to external interrupts.

The RT/68MR is provided in an MCM6830D mask-programmed ROM, and is available from stock, priced at \$55.00 postpaid.

6800 Disk System

The MSI FD-8 Floppy Disk Memory System interfaces to any microcomputer system via a single PIA chip. The FD-8 uses GSI disk drives with each drive housed in its own cabinet complete with power supply. The disk controller board is contained in the same cabinet as the No. 1 drive and communicates to the microcomputer system by means of a small ribbon cable. Up to four disk drives connect to the No. 1 drive by means of a parallel cable. Each additional drive is contained in its own matching cabinet with power supply.

The FD-8 controller board contains a sector buffer. Approximately 3K of RAM is contained on the controller board itself, which allows information to be transferred from controller to disk completely independently from processor speed. In order to execute a transfer, information is first transferred from the main computer memory to the sector buffer RAM, the desired track and sector is then found, and a control bit is used to start the transfer of information. Approximately 512 bytes of memory in the microcomputer system are necessary to house the disk driver and FDOS bootstrap routine. The complete disk driver subroutines occupy approximately 1K of memory and are loaded by the bootstrap.

The interface to the microcomputer system is by means of a single PIA chip. One half of the chip is utilized as an eight bit bi-directional port for data flow and status information. The second half of the PIA is used as an output control port. The MSI PIA-1 parallel interface card is all that is needed for SWTP 6800 systems. An 88-4PIO or 3P+S interface card will handle the interfacing job for Altair and IMSAI 8080 systems.

The FD-8 uses hard sectoring and writes 256 data bytes per sector. Available software includes all of the

driver subroutines and MINI-DOS routines which allow the user to read or write any desired number of sectors to and from a desired starting location on the disk and any desired location in computer memory. On write operations, we read back after write to check for errors in order to insure correct transfer of the information. Error routines are part of the software and indicate to the user the nature of the error, should one occur. Error messages include "Disk Not Ready," "Fault," "Write Protect," "Track or Sector Identification," and "Checksum." The FD-8 writes preambles, postambles, track and sector identification, and checksum on each sector of the disk.

The MSI FDOS Floppy Disk Operating System is available for 6800 based systems. A complete description of the features of our operating system is given in this catalogue. (See page 8) For 8080 based systems, our disk driver routines and MINI-DOS may be integrated with BASIC via user-defined subroutines.

The FD-8 System is furnished with complete manuals and documentation including schematics, assembly manual, trouble-shooting and waveform analysis guide, operating instructions, complete source listings of disk driver, diagnostics, and MINI-DOS software routines. The FD-8 System includes disk drive, controller board, power supply, and cabinet. Driver routines, diagnostics, and MINI-DOS are furnished on tape cassette with the system.

Midwest Scientific Instruments

Prices :	Kit	Ass'd
Single Drive FD-8 Floppy Drive System	\$1375.	\$1625.
Dual Drive " " " "	2325.	2675.
MSI Disk Basic (contains FDOS with named files)	-	80.
I.C. Socket Kit	40.	-
Additional FD-8 Disk Drive	-	1075.

Prices F.O.B. Toronto. Delivery from stock to 30 days.

GraphicAdd

FOR

Sol AND VDM

GraphicAdd will give your Sol computer or VDM-1 Video Display Module graphic capability - with a matrix resolution of 128H x 48V.

It's ideal for: plotting graphs in BASIC
video games and animation
mixing text and graphics

Sol ready-to-load software package (on Sol cassette and PT) includes:
Graphics Driver
BASIC 5 links
LIFE animation program
...all with commented 8080 source and sample programs.

BASIC 5 links: Provide linkage from BASIC 5 (or any BASIC with machine language CALL and parameter passing in D,E registers) to Graphics Driver. This allows X,Y plotting on to the screen display directly from BASIC.

LIFE: A full-blown animation program using Conway's rules of Life complete with generation and population readout. LIFE starts up with user instructions and enters to a Super-Doodle Inoculator that can draw points, lines (including diagonals), and zoom repeat as well. Inoculate the display with cell cultures, then GO LIFE! LIFE is assembled at 0 - 68B.



Demonstration programs include plotting of spirals, flowers, etc. for BASIC, and a machine language screen 'painter'.

HARDWARE SPECIFICATIONS

GraphicAdd is a piggyback PC board which mounts directly on to the VDM and Sol boards (so it doesn't occupy valuable S-100 bus space). No modifications that affect normal circuit operation are necessary. GraphicAdd works by replacing half of the inverse video character set by bit-mapped graphic cells. Options for enabling of graphics mode include fixed graphics, switch-selectable graphics, and even programmable graphics mode!

SOFTWARE SPECIFICATIONS

Graphics Driver: A utility routine which accepts H and V coordinates of any cell and modifies or examines the specified cell. Subroutines can set cells to WHITE, reset cells to BLACK, toggle cell states, or set graphics mode. Resides in Sol at CB00 - CB7A.

PRICING

GraphicAdd Kit with hi-quality, plated-thru, solder-masked PC board, and all parts, including sockets and factory-prime ICs. Documentation includes manual and fully commented source listings for all software described above - on Sol cassette (CUTS), and paper tape....
Introductory price \$50.

DELIVERY: stock to 30 days

*Sol and VDM are registered trademarks of Processor Technology Corp., Emeryville, Calif.

KEA Micro Design

**FIRST CANADIAN
COMPUTER STORE LTD**

44 Eglinton Avenue West
Toronto, Ontario M4R 1A1

Please send me:

_____	SOL Music system kit	@ \$30.00	= \$	_____
_____	RT/68M Multitasking System	@ \$55.00	= \$	_____
_____	SOL & VDM Graphic-Add kit	@ \$50.00	= \$	_____
_____	_____	@ \$	= \$	_____
_____	_____	@ \$	= \$	_____
	Ontario residents add 7% P.S.T.		= \$	_____
	I enclose full payment of		= \$	_____

Name _____
Address _____

NORTHSTAR COMPUTERS INC. PRODUCTS ADDED TO CATALOG :

Single Minifloppy disk drive system	Kit: \$809.	Ass'd: \$919.
Extra drive	489.	-
Power supply	45.	-
Cabinet for single drive	45.	-
Dual Minifloppy disk drive system	1388.	1498.

**FIRST CANADIAN
COMPUTER STORE LTD**

44 Eglinton Avenue West
Toronto, Ontario M4R 1A1



PLACE
STAMP
HERE

3EK-222

PDF Created by David Crawford, www.the-crawfords.com